

Stamp Applications no. 4 (June '95):

## High-Precision Measurement Made Easy With New 12-bit Analog-to-Digital Converter

Using the LTC1298, by Scott Edwards


MANY popular applications for the Stamp include analog measurement, either using the built-in Pot (resistance measurement) command or an external analog-to-digital converter (ADC). These measurements are limited to eight-bit resolution, meaning that a 5-volt full-scale measurement would be broken into units of  $5/256 = 19.5$  millivolts (mV).

That sounds pretty good until you apply it to a real-world sensor. Take the LM34 and LM35 temperature sensors as an example. They output a voltage proportional to the ambient temperature in degrees Fahrenheit (LM34) or Centigrade (LM35). A 1-degree change in temperature causes a 10-mV change in the sensor's output voltage. An eight-bit conversion

gives lousy 2-degree resolution. By reducing the ADC's range, or amplifying the sensor signal, you can improve resolution at the expense of more components and a less-general design.

The easy way out is to switch to an ADC with 10- or 12-bit resolution. Until recently, that hasn't been a decision to make lightly, since more bits = more bucks. However, the new LTC1298 12-bit ADC is reasonably priced at less than \$10, and gives your Stamp projects two channels of 1.22-millivolt resolution data. It's available in a Stamp-friendly 8-pin DIP, and draws about 250 microamps (uA) of current.

The figure shows how to connect the LTC1298 to the Stamp, and the listing supplies the necessary driver code.



*Pinout and connection details for the LTC1298.*

If you have used other synchronous serial devices with the Stamp, such as EEPROMs, other ADCs or the DS1620 thermometer described in a previous column, there are no surprises here. We have tied the LTC1298's data input and output together to take advantage of the Stamp's ability to switch data directions on the fly. The resistor limits the current flowing between the Stamp I/O pin and the 1298's data output in case a programming error or other fault causes a "bus conflict." This happens when both pins are in output mode and in opposite states (1 vs. 0). Without the resistor, such a conflict would cause large currents to flow between pins, possibly damaging the Stamp and/or ADC.

You may have noticed that the LTC1298 has no voltage-reference (Vref) pin. The voltage reference is what an ADC compares its analog input voltage to. When the analog voltage is equal to the reference voltage, the ADC outputs its maximum measurement value; 4095 in this case. Smaller input voltages result in proportionally smaller output values. For example, an input of 1/10th the reference voltage would produce an output value of 409.

The LTC1298's voltage reference is internally connected to the power supply at pin 8. This means that a full-scale reading of 4095 will occur when the input voltage is equal to the power-supply voltage, nominally 5 volts. Notice the weasel word "nominally," meaning "in name only." The actual voltage at the +5-volt rail of the full-size (pre-BS1-IC) Stamp with the LM2936 regulator can be 4.9 to 5.1 volts initially, and can vary by 30 mV.

In some applications you'll need a calibration step to compensate for the supply voltage. Suppose the LTC1298 is looking at a source of 2.00 volts. If the supply is 4.90 volts, the LTC1298 will measure  $(2.00/4.90) * 4095 = 1671$ . If the supply is at the other extreme, 5.10 volts, the LTC1298 will measure  $(2.00/5.10) * 4095 = 1606$ .

How about that 30-millivolt deviation in regulator performance, which cannot be calibrated away? If calibration makes it seem as though the LTC1298 is getting a 5.000-volt reference, a 30-millivolt variation means that

the voltage would vary 15 millivolts high or low. Using the 2.00-volt example, the LTC1298 measurements can range from  $(2.00/4.985) * 4095 = 1643$  to  $(2.00/5.015) * 4095 = 1633$ .

The bottom line is that the measurements you make with the LTC1298 will be only as good as the stability of your +5-volt supply.

I suppose the reason for leaving off a separate voltage-reference pin was to make room for the chip's second analog input. The LTC1298 can treat its two inputs as either separate ADC channels, or as a single, differential channel. A differential ADC is one that measures the voltage difference between its inputs, rather than the voltage between one input and ground.

A final feature of the LTC1298 is the sample-and-hold capability. At the instant your program requests data, the ADC grabs and stores the input voltage level in an internal capacitor. It measures this stored voltage, not the actual input voltage.

By measuring this voltage snapshot, the LTC1298 avoids the errors that can occur when an ADC tries to measure a changing voltage. Without going into the gory details, most common ADCs are *successive approximation* types. That means that they zero in on a voltage measurement by comparing a guess to the actual voltage, then determining whether the actual is higher or lower. They formulate a new guess and try again. This game becomes very difficult if the voltage is constantly changing!

ADCs that aren't equipped with sample-and-hold circuitry should not be used to measure noisy or fast-changing voltages. The LTC1298 has no such restriction.

The program listing is thoroughly commented, so I won't waste ink by repeating that stuff here. Instead, I want discuss a subtlety that trips up many Stamp users: the difference between an I/O pin *number* and a pin *variable*.

Look at the beginning of the listing, under ADC Interface Pins. We've assigned names (SYMBOLs) to each of the pins that the Stamp uses to talk to the LTC1298. CS is 0, CLK is 1, and DIO (data in/out) is 2. DIO actually has two SYMBOLs: DIO\_n (2) and DIO\_p (pin2). Why?

PBASIC refers to the I/O pins in one of two ways; as numbers (0 through 7) or as bit

variables with preassigned names (pin0 through pin7). The following commands use pin numbers:

*High, Low, Toggle, Input, Output, Reverse, Pot, Pulsin, Pulsout, PWM, Serin, Serout, Sound Take High* for example. The following are valid PBASIC commands:

```
High 3      ' Make pin 3 output high.
High b2     ' Make the pin number (0-7)
            ' contained in b2 output high.
High bit7   ' Make pin 0 high if bit7 = 0;
            ' make pin 1 high if bit7 = 1.
```

You can specify the pin as either a constant like "3" or as a variable like b2 or bit7.

On the other hand, the math and logic instructions look at the pins as bit variables with a value of 0 or 1. Suppose pin 1 is set up as an input, and the following instruction executes:

```
Let b2 = b2 + pin1 ' Add pin 1 to b2.
```

If there's a 0 at pin 1, 0 is added to b2. If there's a 1 at pin 1, 1 is added to b2. This example points out why we sometimes refer to pins by their numbers, and sometimes by their variable names. Would the following instruction have the same result as the one above?

```
Let b2 = b2 + 1    ' Add 1 to b2.
```

Nope. We have to make the distinction between *1* and *pin1*.

Back to the question I originally posed: Why assign two SYMBOLS for the DIO pin? The

reason is that this pin is used with High, which requires a pin number, and also with a Let expression, which requires a variable name. The listing's convention for this is to add the ending "\_n" for the number and "\_p" for the pin-variable name.

The common bug that arises from the two ways of referring to pins looks like this:

```
High pin3    ' Usually a bug!
```

The programmer believes that he or she is making pin 3 output high. PBASIC interprets this line to mean, "if pin3 = 0 then make pin 0 high; if pin3 = 1 then make pin 1 high." Probably not what the programmer had in mind.

**Conclusion.** The LTC1298 is a Stamp-friendly peripheral for precision sensing applications. It's available from Digi-Key (800-344-4539) for \$8.89 in single quantity as part number LTC1298CN8-ND (8-pin DIP) or LTC1298CS8-ND (surface-mount). Be sure to request a data sheet or order the data book (9210B-ND, \$9.95) when you order.

For a copy of the program listing (plus a thermostat programming example and PIC assembly language source code) on disk, a sample LTC1298, and LTC1298documentation, you may order the LTC1298App Kit from Scott Edwards Electronics, 964 Cactus Wren Lane, Sierra Vista, AZ 85635; phone, 520-459-4802; fax 520-459-0623. Price is \$25 postpaid. Visa, Mastercard, American Express, checks, and qualified purchase orders accepted.

```
' Program: LTC1298 (LTC1298 analog-to-digital converter)
' The LTC1298 is a 12-bit, two-channel ADC. Its high resolution, low
' supply current, low cost, and built-in sample/hold feature make it a
' great companion for the Stamp in sensor and data-logging applications.
' With its 12-bit resolution, the LTC1298 can measure tiny changes in
' input voltage; 1.22 millivolts (5-volt reference/4096).
' =====
'
'                 ADC Interface Pins
' =====
' The 1298 uses a four-pin interface, consisting of chip-select, clock,
' data input, and data output. In this application, we tie the data lines
' together with a 1k resistor and connect the Stamp pin designated DIO
' to the data-in side of the resistor. The resistor limits the current
' flowing between DIO and the 1298's data out in case a programming error
' or other fault causes a "bus conflict." This happens when both pins are
' in output mode and in opposite states (1 vs 0). Without the resistor,
' such a conflict would cause large currents to flow between pins,
' possibly damaging the Stamp and/or ADC.
SYMBOL   CS = 0           ' Chip select; 0 = active.
SYMBOL   CLK = 1          ' Clock to ADC; out on rising, in on falling edge.
SYMBOL   DIO_n = 2        ' Pin _number_ of data input/output.
SYMBOL   DIO_p = pin2     ' Variable_name_ of data input/output.
SYMBOL   ADbits = b1      ' Counter variable for serial bit reception.
SYMBOL   AD = w1          ' 12-bit ADC conversion result.
' =====
'
'                 ADC Setup Bits
' =====
' The 1298 has two modes. As a single-ended ADC, it measures the
' voltage at one of its inputs with respect to ground. As a differential
' ADC, it measures the difference in voltage between the two inputs.
' The sglDif bit determines the mode; 1 = single-ended, 0 = differential.
' When the 1298 is single-ended, the oddSign bit selects the active input
' channel; 0 = channel 0 (pin 2), 1 = channel 1 (pin 3).
' When the 1298 is differential, the oddSign bit selects the polarity
' between the two inputs; 0 = channel 0 is +, 1 = channel 1 is +.
' The msbf bit determines whether clock cycles _after_ the 12 data bits
' have been sent will send 0s (msbf = 1) or a least-significant-bit-first
' copy of the data (msbf = 0). This program doesn't continue clocking after
' the data has been obtained, so this bit doesn't matter.
' You probably won't need to change the basic mode (single/differential)
' or the format of the post-data bits while the program is running, so
' these are assigned as constants. You probably will want to be able to
' change channels, so oddSign (the channel selector) is a bit variable.
SYMBOL   sglDif = 1       ' Single-ended, two-channel mode.
SYMBOL   msbf = 1         ' Output 0s after data transfer is complete.
SYMBOL   oddSign = bit0   ' Program writes channel # to this bit.
```

```

' =====
'
'                               Demo Program
' =====
' This program demonstrates the LTC1298 by alternately sampling the two
' input channels and presenting the results on the PC screen using Debug.
high CS           ' Deactivate the ADC to begin.
Again:           ' Main loop.
  For oddSign = 0 to 1 ' Toggle between input channels.
    gosub Convert      ' Get data from ADC.
    debug "ch ",#oddSign,":",#AD,cr ' Show the data on PC screen.
    pause 500         ' Wait a half second.
  next
goto Again       ' Endless loop.

' =====
'
'                               ADC Subroutine
' =====
' Here's where the conversion occurs. The Stamp first sends the setup
' bits to the 1298, then clocks in one null bit (a dummy bit that always
' reads 0) followed by the conversion data.
Convert:
  low CLK           ' Low clock--output on rising edge.
  high DIO_n       ' Switch DIO to output high (start bit).
  low CS           ' Activate the 1298.
  pulsout CLK,5    ' Send start bit.
  let DIO_p = sglDif ' First setup bit.
  pulsout CLK,5    ' Send bit.
  let DIO_p = oddSign ' Second setup bit.
  pulsout CLK,5    ' Send bit.
  let DIO_p = msbf  ' Final setup bit.
  pulsout CLK,5    ' Send bit.
  input DIO_n      ' Get ready for input from DIO.
  let AD = 0       ' Clear old ADC result.
  for ADbits = 1 to 13 ' Get null bit + 12 data bits.
    let AD = AD*2+DIO_p ' Shift AD left, add new data bit.
    pulsout CLK,5      ' Clock next data bit in.
  next
  high CS          ' Turn off the ADC
return            ' Return to program.

```